

Advanced Topics from Scientific Computing

TU Berlin Winter 2023/24

Notebook 11



Jürgen Fuhrmann

Grid creation and visualization

This notebook shows how to perform grid creation and visualization with the assistance of the packages [ExtendableGrids.jl](#) and [SimplexGridFactory.jl](#) using [Triangulate.jl](#), [TetGen.jl](#) and [Gmsh.jl](#). Visualization in this notebook is done using the [GridVisualize.jl](#) package which itself relies on various backends.

These packages will be used later on by [VoronoiFVM.jl](#) for solving partial differential equations.

```
1 begin
2     using SimplexGridFactory
3     using ExtendableGrids
4     using Triangulate
5     using TetGen
6     using GridVisualize
7     using PlutoVista
8     using PlutoUI
9     import CairoMakie
10    default_plotter!(CairoMakie)
11    CairoMakie.activate!(type="svg")
12 end
```

1D grids

1D grids are created just from arrays of monotonically increasing coordinates using the [simplexgrid](#) method.


`X1 = 0.0:0.05:1.0`

```
1 X1 = range(0, 1; length = 21)
```

```
g1 = ExtendableGrids.ExtendableGrid{Float64, Int32};
    dim: 1 nodes: 21 cells: 20 bfaces: 2
```

```
1 g1 = simplexgrid(X1)
```

We can plot a grid with a method from GridVisualize.jl



```
1 gridplot(g1; resolution = (500, 150), legend = :rt)
```

We see some additional information:

- cellregion: each grid cell (interval, triangle, tetrahedron) as an integer region marker attached
- bface region: boundary faces (points, lines, triangles) have an integer boundary region marker attached

We can also have a look into the grid structure:

```
Dict{XCoordinates => [0.0, 0.05, 0.1, 0.15, 0.2, 0.25, 0.3, 0.35, 0.4, more ,1.0], Coor
```

```
1 g1.components
```

Components can be accessed via []. In fact the keys in the dictionary of components are types.

```
1x21 Matrix{Float64}:
 0.0 0.05 0.1 0.15 0.2 0.25 0.3 0.35 ... 0.7 0.75 0.8 0.85 0.9 0.95 1.0
```

```
1 g1[Coordinates]
```

```
2x20 Matrix{Int32}:
 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20
 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21
```

```
1 g1[CellNodes]
```

Modifying region markers


The simplexgrid method provides a default distribution of markers, but we would like to be able to change them. This can be done by putting masks on cells or faces (points in 1D):

```
g2 = ExtendableGrids.ExtendableGrid{Float64, Int32};
    dim: 1 nodes: 21 cells: 20 bfaces: 2
```

```
1 g2 = deepcopy(g1)
```

```
1 cellmask!(g2, [0.0], [0.5], 2);
```

```
1 bfacemask!(g2, [0.5], [0.5], 3);
```



```
1 gridplot(g2; resolution = (500, 150), legend = :rt)
```

Creating locally refined grids

For this purpose, we just need to create arrays with the corresponding coordinate values. This can be done programmatically.

Two support methods are provided for this purpose.

0.1

```
1 begin
2   hmin = 0.01;
3   hmax = 0.1;
4 end
```

The `geomspace` method creates an array such that the smallest interval size is `hmin` and the largest interval size is not larger but close to `hmax`, and the interval sizes constitute a geometric sequence.

`X2L =`

```
[0.0, 0.0931551, 0.170501, 0.234722, 0.288044, 0.332316, 0.369076, 0.399597, 0.424939, 0.
```

```
1 X2L = geomspace(0, 0.5, hmax, hmin)
```

`DX2 =`

```
[0.0931551, 0.0773463, 0.0642203, 0.0533218, 0.0442729, 0.0367596, 0.0305213, 0.0253417,
```

```
1 DX2 = X2L[2:end] .- X2L[1:(end - 1)]
```

```
[1.20439, 1.20439, 1.20439, 1.20439, 1.20439, 1.20439, 1.20439, 1.20439, 1.20439, 1.20439,
```

```
1 DX2[1:(end - 1)] ./ DX2[2:end]
```

X2R =

```
[0.5, 0.51, 0.522044, 0.536549, 0.55402, 0.575061, 0.600403, 0.630924, 0.667684, 0.711956
```


```
1 X2R = geomspace(0.5, 1, hmin, hmax)
```

We can glue these arrays together and create a grid from them:

X2 =

```
[0.0, 0.0931551, 0.170501, 0.234722, 0.288044, 0.332316, 0.369076, 0.399597, 0.424939, 0.
```

```
1 X2 = glue(X2L, X2R)
```



```
1 gridplot(simplexgrid(X2); resolution = (500, 150), legend = :rt)
```

Plotting functions

We assume that functions can be represented by their node values and plotted via their piecewise linear interpolants. E.g. they could come from some simulation.

```
g1d2 = ExtendableGrids.ExtendableGrid{Float64, Int32};
dim: 1 nodes: 201 cells: 200 bfaces: 2
```

```
1 g1d2 = simplexgrid(range(-10, 10; length = 201))
```

fsin =

```
[0.544021, 0.457536, 0.366479, 0.271761, 0.174327, 0.0751511, -0.0247754, -0.124454, -0.2
```

```
1 fsin = map(sin, g1d2)
```

fcos =


```
[-0.839072, -0.889191, -0.930426, -0.962365, -0.984688, -0.997172, -0.999693, -0.992225,
```

```
1 fcos = map(cos, g1d2)
```

fsinh =

```
[-3.62686, -3.55234, -3.47923, -3.40752, -3.33718, -3.26816, -3.20046, -3.13403, -3.06886
```

```
1 fsinh = map(x -> sinh(0.2 * x), g1d2)
```



```

1 let
2   vis = GridVisualizer(; resolution = (600, 300), legend = :lt )
3
4   scalarplot!(vis, g1d2, fsinh; label = "sinh", markershape = :dtriangle, color =
: red, markevery = 5, clear = false)
5
6   scalarplot!(vis, g1d2, fcos; label = "cos", markershape = :xcross, color =
: green, linestyle = :dash, clear = false,
7     markevery = 20)
8
9   scalarplot!(vis, g1d2, fsin; label = "sin", markershape = :none, color = :blue,
linestyle = :dot, clear = false, markevery = 20)
10
11  reveal(vis)
12 end

```


2D grids

Tensor product grids

For 2D tensor product grids, we can again use the `simplexgrid` method and apply the mask methods for modifying cell and boundary region markers.

```
ExtendableGrids.ExtendableGrid{Float64, Int32};
dim: 2 nodes: 567 cells: 1040 bfaces: 92
```

```
1 begin
2   g2d1 = simplexgrid(X1, X2)
3   cellmask!(g2d1, [0.0, 0.0], [0.5, 0.5], 2)
4   cellmask!(g2d1, [0.5, 0.5], [1.0, 1.0], 3)
5   bfacemask!(g2d1, [0.0, 0.0], [0.0, 0.5], 5)
6 end
```



```
1 gridplot(g2d1; resolution = (600, 400), linewidth = 0.5)
```

To interact with the plot when using the PlutoVista backend, you can use the mouse wheel or double touch to zoom, "shift-mouse" to pan, and "alt-mouse-left" or "ctrl-mouse-left" to reset.

We can also have a look into the components of a 2D grid:

```
Dict{XCoordinates => [0.0, 0.05, 0.1, 0.15, 0.2, 0.25, 0.3, 0.35, 0.4, more ,1.0], Coor
```

```
1 g2d1.components
```

Unstructured grids

For the triangulation of unstructured grids, we use the mesh generator Triangle via the [Triangulate.jl](#) and [SimplexGridFactory.jl](#) packages.


The later package exports the `SimplexGridBuilder` which shall help to simplify the creation of the input for `Triangulate`.

```
builder2 =
```

```
SimplexGridBuilder(Triangulate, 3, 1, 1.0, 1.0e-12, [1, 2, 3], [[1, 2], [2, 3], [3, 1]], f
```

```
1 builder2 = let
2     b = SimplexGridBuilder(; Generator = Triangulate)
3     p1 = point!(b, 0, 0)
4     p2 = point!(b, 1, 0)
5     p3 = point!(b, 1, 1)
6
7     # Specify outer boundary
8     facetregion!(b, 1)
9     facet!(b, p1, p2)
10    facetregion!(b, 2)
11    facet!(b, p2, p3)
12    facetregion!(b, 3)
13    facet!(b, p3, p1)
14
15    cellregion!(b, 1)
16    regionpoint!(b, 0.75, 0.25)
17
18    options!(b; maxvolume = 0.01)
19    b
20 end
```

We can plot the current state of the builder (in the moment this works only with PyPlot):




```
1 builderplot(builder2; Plotter = CairoMakie)
```

Found `resolution` in the theme when creating a `Scene`. The `resolution` keyword for `Scene`s and `Figure`s has been deprecated. Use `Figure(; size = ...)` or `Scene(; size = ...)` instead, which better reflects that this is a unitless size and not a pixel resolution. The key could also come from `set_theme!` calls or related theming functions.

```
grid2d2 = ExtendableGrids.ExtendableGrid{Float64, Int32};
dim: 2 nodes: 449 cells: 793 bfaces: 103
```

```
1 grid2d2 = simplexgrid(builder2; maxvolume = 0.001)
```



```
1 gridplot(grid2d2; resolution = (400, 300), linewidth = 0.5)
```


More complicated grids

More complicated grids include:

- local refinement
- interior boundaries
- different region markers
- holes

The particular way to describe these things is due to Jonathan Shewchuk and his mesh generator [Triangle](#) via its Julia wrapper package [Triangulate.jl](#).

Local refinement

```
refinement_center = [0.8, 0.2]  
1 refinement_center = [0.8, 0.2]
```

For local refinement, we define a function, which is able to tell if a triangle is to be refined ("unsuitable") or can be kept as it is.

The function measures the distance between the refinement center and the triangle barycenter. We require that the area increases with the distance from the refinement center.

```
1 function unsuitable(x1, y1, x2, y2, x3, y3, area)  
2     bary_x = (x1 + x2 + x3) / 3.0  
3     bary_y = (y1 + y2 + y3) / 3.0  
4     dx = bary_x - refinement_center[1]  
5     dy = bary_y - refinement_center[2]  
6     qdist = dx^2 + dy^2  
7     area > 0.1 * max(1.0e-2, qdist)  
8 end;
```

Interior boundaries

Interior boundaries are described in a similar as exterior ones - just by facets connecting points.


Subregions

Subregions are defined as regions surrounded by interior boundaries. By placing a "region point" into such a region and specifying a "region number", we can set the cell region marker for all triangles created in the subregion.

Holes

Holes are defined in a similar way as subregions, but a "hole point" is placed into the place which shall become the hole.

```
1 builder3 = let
2     b = SimplexGridBuilder(; Generator = Triangulate, tol = 1.0e-10)
3
4     # Specify points
5     p1 = point!(b, 0, 0)
6     p2 = point!(b, 1, 0)
7     p3 = point!(b, 1, 1)
8     p4 = point!(b, 0, 0.7)
9
10    # Specify outer boundary
11    facetregion!(b, 1)
12    facet!(b, p1, p2)
13    facetregion!(b, 2)
14    facet!(b, p2, p3)
15    facetregion!(b, 3)
16    facet!(b, p3, p4)
17    facetregion!(b, 4)
18    facet!(b, p1, p4)
19
20    # Activate unsuitable callback
21    options!(b; unsuitable = unsuitable)
22
23    # Specify interior boundary
24    facetregion!(b, 5)
25    facet!(b, p1, p3)
26
27    # Coarse elements in upper left region #1
28    cellregion!(b, 1)
29    maxvolume!(b, 0.1)
30    regionpoint!(b, 0.1, 0.5)
31
32    # Fine elements in lower right region #2
33    cellregion!(b, 2)
34    maxvolume!(b, 0.01)
35    regionpoint!(b, 0.9, 0.5)
36
37    # Hole
38    hp1 = point!(b, 0.4, 0.1)
39    hp2 = point!(b, 0.6, 0.1)
40    hp3 = point!(b, 0.5, 0.3)
41    holepoint!(b, 0.5, 0.2)
42    facetregion!(b, 6)
43    facet!(b, hp1, hp2)
44    facet!(b, hp2, hp3)
45    facet!(b, hp3, hp1)
46
47    b
48 end;
```




```
1 builderplot(builder3; Plotter = CairoMakie)
```

Found `resolution` in the theme when creating a `Scene`. The `resolution` keyword for `Scene`s and `Figure`s has been deprecated. Use `Figure(; size = ...)` or `Scene(; size = ...)` instead, which better reflects that this is a unitless size and not a pixel resolution. The key could also come from `set_theme!` calls or related theming functions.

Create a simplex grid from the builder

```
grid2d3 = ExtendableGrids.ExtendableGrid{Float64, Int32};  
dim: 2 nodes: 117 cells: 199 bfaces: 47
```

```
1 grid2d3 = simplexgrid(builder3)
```



```
1 gridplot(grid2d3; resolution = (400, 400))
```

Plotting of functions

Functions defined on the nodes of a triangular grid can be seen as piecewise linear functions from the P_1 finite element space defined by the triangulation.

`fsin2 =`


```
[0.0, 0.0, 0.841471, 0.38939, 0.420735, 0.631103, 0.603703, 0.467345, 0.720622, 0.736287,
```

```
1 fsin2 = map((x, y) -> sin(x) * y, grid2d2)
```


`fsin3 =`

```
[0.0, 0.0, 0.841471, 0.0, 0.0399334, 0.0599, 0.14776, 0.0, 0.122412, 0.0, 0.0, 0.38939, 0
```

```
1 fsin3 = map((x, y) -> sin(y) * x, grid2d3)
```



```
1 scalarplot(grid2d2, fsin2; label = "grid2d2")
```



```
1 scalarplot(grid2d3, fsin3; label = "grid2d3", colormap = :spring, isolines = 10)
```

3D Grids

Tensor product grids

Please note that "masking" is not yet implemented. Furthermore, PyPlot visualization is slow, with GLMakie or PlutoVista it is way faster.

```
X3 = 0.0:1.01:10.1
```

```
1 X3 = range(0.0, 10.1; length = 11)
```

```
grid3d1 = ExtendableGrids.ExtendableGrid{Float64, Int32};  
dim: 3 nodes: 1331 cells: 6000 bfaces: 1200
```


```
1 grid3d1 = simplexgrid(X3, X3, X3)
```

func3 =

```
[0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, -0.0, -0.0, -0.0, -0.0, 0.0, 0.0, 0.0, 0.0, 0.0,
```

```
1 func3 = map((x, y, z) -> sin(x / 2) * cos(y / 2) * z / 10, grid3d1)
```


p3dg =



```
1 p3dg = GridVisualizer(; dim = 3, resolution = (400, 400), Plotter=PlutoVista)
```

```
1 gridplot!(p3dg, grid3d1; zplanes = [zplane], yplanes = [yplane], xplanes = [xplane], show = true)
```


p3ds =



```
1 p3ds = GridVisualizer(; dim = 3, resolution = (400, 400), Plotter=PlutoVista)
```

```
1
2 scalarplot!(p3ds, grid3d1, func3; zplanes = [zplane], yplanes = [yplane], xplanes =
  [xplane], levels = [flevel], colormap = :spring,
3           resolution = (200, 200), show = true, levelalpha = 0.5, outlinealpha =
  0.1)
```

f= -0.049773893527225145

x= 10.1

y= 10.1

z= 10.1

mean (generic function with 1 method)


```
1 mean(x) = sum(x) / length(x)
```

```
extruded_grid = ExtendableGrids.ExtendableGrid{Float64, Int32};
                dim: 3 nodes: 1287 cells: 5970 bfaces: 1338
```

```
1 extruded_grid=simplexgrid(grid2d3, 0:0.1:1)
```

16788756 allocations during ExtendableGrids.ITEMTYPE_CELL volume calculation

p3dx =



```
1 p3dx = GridVisualizer(; dim = 3, resolution = (400, 400), Plotter=PlutoVista)
```

```
1 gridplot!(p3dx,extruded_grid, Plotter=PlutoVista,zplanes=[zplane2],show=true)
```

z= 1.0

Unstructured grids

The SimplexGridBuilder API supports creation of three-dimensional grids in way very similar to the 2D case. Just define points with three coordinates and planar (!) facets with at least three points to describe the geometry.

The backend for mesh generation in this case is the [TetGen](#) mesh generator by Hang Si from WIAS Berlin and its Julia wrapper [TetGen.jl](#).

```

1 builder3d = let
2     b = SimplexGridBuilder(; Generator = TetGen)
3
4     p1 = point!(b, 0, 0, 0)
5     p2 = point!(b, 1, 0, 0)
6     p3 = point!(b, 1, 1, 0)
7     p4 = point!(b, 0, 1, 0)
8     p5 = point!(b, 0, 0, 1)
9     p6 = point!(b, 1, 0, 1)
10    p7 = point!(b, 1, 1, 1)
11    p8 = point!(b, 0, 1, 1)
12
13    facetregion!(b, 1)
14    facet!(b, p1, p2, p3, p4)
15    facetregion!(b, 2)
16    facet!(b, p5, p6, p7, p8)
17    facetregion!(b, 3)
18    facet!(b, p1, p2, p6, p5)
19    facetregion!(b, 4)
20    facet!(b, p2, p3, p7, p6)
21    facetregion!(b, 5)
22    facet!(b, p3, p4, p8, p7)
23    facetregion!(b, 6)
24    facet!(b, p4, p1, p5, p8)
25
26    hp1 = point!(b, 0.4, 0.4, 0.4)
27    hp2 = point!(b, 0.6, 0.4, 0.4)
28    hp3 = point!(b, 0.6, 0.6, 0.4)
29    hp4 = point!(b, 0.4, 0.6, 0.4)
30    hp5 = point!(b, 0.4, 0.4, 0.6)
31    hp6 = point!(b, 0.6, 0.4, 0.6)
32    hp7 = point!(b, 0.6, 0.6, 0.6)
33    hp8 = point!(b, 0.4, 0.6, 0.6)
34
35    facetregion!(b, 7)
36    facet!(b, hp1, hp2, hp3, hp4)
37    facet!(b, hp5, hp6, hp7, hp8)
38    facet!(b, hp1, hp2, hp6, hp5)
39    facet!(b, hp2, hp3, hp7, hp6)
40    facet!(b, hp3, hp4, hp8, hp7)
41    facet!(b, hp4, hp1, hp5, hp8)
42    holepoint!(b, 0.5, 0.5, 0.5)
43
44    b
45 end;

```

```


grid3d2 = ExtendableGrids.ExtendableGrid{Float64, Int32};
          dim: 3 nodes: 4650 cells: 21311 bfaces: 4890

```

```

1 grid3d2 = simplexgrid(builder3d; maxvolume = 0.0001)

```



```
1 gridplot(grid3d2; zplane = 0.5, azimuth = 20, elevation = 20, linewidth = 0.5, outlinealpha = 0.3, Plotter=PlutoVista)
```

Gmsh examples

```
1 using Gmsh
```


Gmsh is an open source 3D finite element mesh generator with a built-in CAD engine and post-processor.

ExtendableGrids.jl now has an extension which allows to interact with Gmsh using its Julia API

Example adopted from the documentation of Ferrite.jl, a finite element solver package.

quarter_ring (generic function with 2 methods)

```
1 function quarter_ring(h=0.05)
2     # Initialize gmsh
3     Gmsh.initialize()
4     gmsh.option.set_number("General.Verbosity", 2)
5
6     # Add the points
7     o = gmsh.model.geo.add_point(0.0, 0.0, 0.0, h)
8     p1 = gmsh.model.geo.add_point(0.5, 0.0, 0.0, h)
9     p2 = gmsh.model.geo.add_point(1.0, 0.0, 0.0, h)
10    p3 = gmsh.model.geo.add_point(0.0, 1.0, 0.0, h)
11    p4 = gmsh.model.geo.add_point(0.0, 0.5, 0.0, h)
12
13    # Add the lines
14    l1 = gmsh.model.geo.add_line(p1, p2)
15    l2 = gmsh.model.geo.add_circle_arc(p2, o, p3)
16    l3 = gmsh.model.geo.add_line(p3, p4)
17    l4 = gmsh.model.geo.add_circle_arc(p4, o, p1)
18
19    # Create the closed curve loop and the surface
20    loop = gmsh.model.geo.add_curve_loop([l1, l2, l3, l4])
21    surf = gmsh.model.geo.add_plane_surface([loop])
22
23    # Synchronize the model
24    gmsh.model.geo.synchronize()
25
26    # Create the physical domains
27    gmsh.model.add_physical_group(1, [l1], -1, "Γ1")
28    gmsh.model.add_physical_group(1, [l2], -1, "Γ2")
29    gmsh.model.add_physical_group(1, [l3], -1, "Γ3")
30    gmsh.model.add_physical_group(1, [l4], -1, "Γ4")
31    gmsh.model.add_physical_group(2, [surf])
32
33    # Generate a 2D mesh
34    gmsh.model.mesh.generate(2)
35
36    grid=simplexgrid_from_gmsh(gmsh.model)
37
38    # Finalize the Gmsh library
39    gmsh.finalize()
40
41    return grid
42 end
```



```
1 gridplot(quarter_ring())
```

Table of Contents

Grid creation and visualization

1D grids

- Modifying region markers
- Creating locally refined grids
- Plotting functions

2D grids

- Tensor product grids
- Unstructured grids
- More complicated grids
 - Local refinement
 - Interior boundaries
 - Subregions
 - Holes
- Plotting of functions

3D Grids

- Tensor product grids
- Unstructured grids

Gmsh examples